
Distribute your application over multiple servers with Horde.
Horde is comprised of Horde.DynamicSupervisor, a distributed supervisor, and Horde.Registry, a distributed registry. Horde is built on top of DeltaCrdt.
Read the full documentation on hexdocs.pm.
There is an introductory blog post and a getting started guide. You can also find me in the Elixir slack channel #horde.
Daniel Azuma gave a great talk at ElixirConf US 2018 where he demonstrated Horde's Supervisor and Registry.
Since Horde is built on CRDTs, it is eventually (as opposed to immediately) consistent, although it does sync its state with its neighbours rather aggressively. Cluster membership in Horde is fully dynamic; nodes can be added and removed at any time and Horde will continue to operate as expected. Horde.DynamicSupervisor also uses a hash ring to limit any possible race conditions to times when cluster membership is changing. 
Horde.Registry and Horde.DynamicSupervisor are both designed to stay as close as possible to the API and behavior of their counterparts in Elixir’s standard library. For most scenarios, they can be used as drop-in replacements with minimal changes required.
Some differences do exist — such as the current lack of support for keys: :duplicate in Horde.Registry — but these divergences occur only when standard library behavior does not translate well to a system that is inherently distributed.
Our goal is to keep these differences to the absolute minimum necessary, while ensuring that Horde remains reliable, consistent, and optimized for distributed environments. See documentation of Horde.DynamicSupervisor.start_link/1 for details.

  
    
  
  Running a single global process


If you simply need to run a single process as a singleton in your cluster, I would encourage you to look at Highlander or HighlanderPG instead, as one of these may fit your use case better.

  
    
  
  1.0 release


Help us get to 1.0, please fill out our very short survey and report any issues you encounter when using Horde.

  
    
  
  Fault tolerance


If a node fails (or otherwise becomes unreachable) then Horde.DynamicSupervisor will redistribute processes among the remaining nodes.
You can choose what to do in the event of a network partition by specifying :distribution_strategy in the options for Horde.DynamicSupervisor.start_link/2. Setting this option to Horde.UniformDistribution (which is the default) distributes processes using a hash mechanism among all reachable nodes. In the event of a network partition, both sides of the partition will continue to operate. Setting it to Horde.UniformQuorumDistribution will operate in the same way, but will shut down if less than half of the cluster is reachable.

  
    
  
  CAP Theorem


Horde is eventually consistent, which means that Horde can guarantee availability and partition tolerancy. Horde cannot guarantee consistency. This means you may end up with duplicate processes in your cluster. Horde does aggressively synchronize between nodes (this is also tunable), but ultimately, depending on the tuning parameters you choose and the quality of the network, there are conditions under which it is possible to have duplicate processes in your cluster. Horde.Registry terminates duplicate processes as soon as they are discovered with a special exit code, so you'll always know when this is happening. See this page in the docs for more details.
NOTE: Since Horde 0.6.0, Horde.DynamicSupervisor ignores the id of a child spec (as Elixir.DynamicSupervisor does), and therefore does not guarantee that each id will be unique in the cluster (as it did pre-0.6.0). If you want to uniquely name your processes in a cluster, use Horde.Registry for this purpose. Having both Horde.DynamicSupervisor and Horde.Registry checking for uniqueness was subject to a race condition where Horde.DynamicSupervisor would choose process A to survive and Horde.Registry would choose process B to survive, resulting in both processes being killed.

  
    
  
  Graceful shutdown


Using Horde.DynamicSupervisor.stop/3 will cause the local supervisor to stop and any processes it was running will be shut down and redistributed to remaining supervisors in the horde. (This should happen automatically if :init.stop() is called).

  
    
  
  Installation


Horde is available in Hex.
The package can be installed by adding horde to your list of dependencies in mix.exs:
def deps do
  [
    {:horde, "~> 0.8.5"}
  ]
end

  
    
  
  Usage


Here is a small taste of Horde's usage. See the full docs at https://hexdocs.pm/horde for more information and examples. There is also an example application at examples/hello_world that you can refer to if you get stuck.
Starting Horde.DynamicSupervisor:
defmodule MyApp.Application do
  use Application
  def start(_type, _args) do
    children = [
      {Horde.DynamicSupervisor, [name: MyApp.DistributedSupervisor, strategy: :one_for_one]}
    ]
    Supervisor.start_link(children, strategy: :one_for_one)
  end
end
Adding a child to the supervisor:
# Add a Task
Horde.DynamicSupervisor.start_child(MyApp.DistributedSupervisor, %{id: :task, start: {Task, :start_link, [:infinity]}})

# Add an Agent
Horde.DynamicSupervisor.start_child(MyApp.DistributedSupervisor, %{id: :agent, start: {Agent, :start_link, [fn -> %{} end]}})

# Add a GenServer: You need a previously defined GenServer to call the one
# liner below.  We have a test ("graceful shutdown") in
# `test/supervisor_test.exs` that exercises and displays that behavior. After
# defined, it would be very similar to this:
Horde.DynamicSupervisor.start_child(MyApp.DistributedSupervisor, %{id: :gen_server, start: {GenServer, :start_link, [DefinedGenServer, {500, pid}]}})
And so on. The public API should be the same as Elixir.DynamicSupervisor (and please open an issue if you find a difference).
Joining supervisors into a single distributed supervisor can be done using Horde.Cluster:
{:ok, supervisor_1} = Horde.DynamicSupervisor.start_link(name: :distributed_supervisor_1, strategy: :one_for_one)
{:ok, supervisor_2} = Horde.DynamicSupervisor.start_link(name: :distributed_supervisor_2, strategy: :one_for_one)
{:ok, supervisor_3} = Horde.DynamicSupervisor.start_link(name: :distributed_supervisor_3, strategy: :one_for_one)

Horde.Cluster.set_members(:distributed_supervisor_1, [:distributed_supervisor_1, :distributed_supervisor_2, :distributed_supervisor_3])
# supervisor_1, supervisor_2 and supervisor_3 will be joined in a single cluster.
Other projects
Useful libraries that use or extend Horde functionalities.

  
    
  
  Horde.Process


An opinionated but configurable means of quickly creating GenServer modules that are intended to be managed and distributed via Horde.
Contributing
Contributions are welcome! Feel free to open an issue if you'd like to discuss a problem or a possible solution. Pull requests are much appreciated.



  

    
Changelog
    


  
    
  
  0.10.0


	Added optional TTL to Horde.DynamicSupervisor's :proxy_operation messages. The Time-to-Live defaults to :infinity for full backwards compatibility. This TTL helps prevent potential issues where messages could loop forever between a set of nodes which disagree on which node should execute the task.
	[BREAKING] Horde.DynamicSupervisor's new :proxy_message_ttl option configures the maximum TTL for proxy messages. It takes an integer denoting the maximum number of hops a message can travel, or the atom :infinity (default). This can be a breaking change: when upgrading do not set this option to an integer. You can explicity set it to :infinity or leave it default. If this is set to an integer, upgraded nodes won't be able to proxy to non-upgrade nodes.


  
    
  
  0.9.1


	Fix race condition in registry when node disconnects
	Pass extra_arguments flag to the ProcessSupervisor
	Updating libring dependency to ~> 1.7. Needed for upgrade to OTP 27. See this PR to libring for details.


  
    
  
  0.9.0


	Bugfixes for scenarios causing Horde to crash. See #266 and #263.
	[BREAKING] The first parameter of Horde.DistributionStrategy.choose_node/2 has changed from the identifier to the full child spec. See #239.
	Use :erpc instead of :rpc. See #265.
	Stop eagerly registering processes in Horde.Registry. This solves #250.
	Add Horde.UniformRandomDistribution process distribution strategy. See #252.


  
    
  
  0.8.7


	Tweak dependency on :telemetry


  
    
  
  0.8.6


	Fix an issue with members: :auto


  
    
  
  0.8.5


	Add support for telemetry version 1.0.0


  
    
  
  0.8.4


	Upgrade to delta_crdt 0.6.0


  
    
  
  0.8.3


	Fix a deadlock that occurred if a process was being restarted while Horde.DynamicSupervisor.start_child/2 was being called. #218
	Respect max_restarts and max_seconds when given as options to Horde.DynamicSupervisor.child_spec/1. #216


  
    
  
  0.8.2


	Bump version of telemetry_poller dependency. #212


  
    
  
  0.8.1


	Horde.Registry.delete_meta/2 has been added to reflect its addition in Elixir.Registry in upcoming release 1.11.0 #208


  
    
  
  0.8.0


	Horde.DynamicSupervisor behaviour in a netsplit has changed. Previously, when a netsplit heals, Horde.DynamicSupervisor would try to clean up any duplicate processes. It no longer does this, leaving that responsibility to Horde.Registry. #196
	Horde.DynamicSupervisor and Horde.Registry now support the option members: :auto to automatically detect other identically-named supervisors or registries. #184
	Horde.DynamicSupervisor now supports the option process_redistribution: :active to rebalance processes actively (aka, when a node joins or leaves the cluster). The default is :passive, which only redistributes processes when a node dies or loses quorum. #164.


  
    
  
  0.7.1


	Use MFA for on_diff instead of anonymous function, avoids passing around functions (which can be error-prone). #167.


  
    
  
  0.7.0


	Horde.Supervisor has been renamed to Horde.DynamicSupervisor.
	Horde.Registry and Horde.Supervisor now follow the api of Elixir.DynamicSupervisor more closely (specifically init/1 callback and module-based Supervisor / Registry). #152.
	Horde.Registry.lookup/2 returns [] instead of :undefined when no match. #145
	child_spec/1 can be overridden in Horde.Registry and Horde.Supervisor #135 #143
	Implement :listeners option for Horde.Registry. #142
	Fix via tuple usage with meta. #139


  
    
  
  0.6.1


	Module-based Horde.Supervisor can override child_spec/1. #135
	Added guides for handling clustering, process state handoff (during deploys), and special considerations for eventual consistency to the documentation.
	Horde.Supervisor now uses libring to distribute processes over nodes. #130
	Horde.Supervisor publishes metrics with :telemetry ([:horde, :supervisor, :supervised_process_count]). #132
	Horde.Supervisor and Horde.Registry now support option delta_crdt_options, which you can use to tune your cluster. Also updated to the most recent DeltaCRDT. #100


  
    
  
  0.6.0


	Horde.Supervisor now behaves more like DynamicSupervisor. #122
	Horde.Registry sends an exit signal to the process that "loses" when a conflict is resolved. #118
	Horde.Registry.register/3 returns {:error, {:already_registered, pid}} when applicable. This improves compatibility with Elixir.Registry. #115
	Adds Horde.Registry.select/2, which works the same as Elixir.Registry.select/2, which will land in Elixir 1.9. #110
	Fixes a bug causing Horde.Supervisor to crash if a child process was restarting when Horde.Supervisor.delete_child/2 was called. #114




  

    
Getting Started
    

This guide shows you how to get started with Horde. Code samples come from the example HelloWorld app in examples/hello_world.

  
    
  
  Erlang clustering


Horde relies on Erlang's built in clustering. To make this work, each node in our cluster needs a unique name, and each node must be started using the same cookie. We can try this out locally like so:
iex --name node1@127.0.0.1 --cookie asdf -S mix
iex --name node2@127.0.0.1 --cookie asdf -S mix
iex --name node3@127.0.0.1 --cookie asdf -S mix

In this example each node has a unique name, and they all share the same cookie. Now we can connect these nodes by running the following code:
Node.connect(:"node2@127.0.0.1")
Run Node.list() to confirm that the nodes are connected.
Horde assumes that you will manage Erlang clustering yourself. There are libraries that will help you do this. Continue reading the getting started guide and when you are getting ready to deploy your application, read about how to set up a cluster.

  
    
  
  Starting Horde.DynamicSupervisor


Horde.DynamicSupervisor is API-compatible with Elixir's DynamicSupervisor. There are extra arguments that you can provide, but the basic recipe stays the same:
defmodule HelloWorld.Application do
  use Application

  def start(_type, _args) do
    children = [
      {Horde.DynamicSupervisor, [name: HelloWorld.HelloSupervisor, strategy: :one_for_one]},
    ]

    opts = [strategy: :one_for_one, name: HelloWorld.Supervisor]
    Supervisor.start_link(children, opts)
  end
end
This is also where you can set additional options for Horde.DynamicSupervisor:
	:members, a list of members (if your cluster will have static membership)
	:distribution_strategy, the distribution strategy (Horde.UniformDistribution is default)
	:delta_crdt_options, for tuning the delta CRDT that underpins Horde

See the documentation for Horde.DynamicSupervisor for more information.

  
    
  
  Starting Horde.Registry


Horde.DynamicSupervisor is spreading your processes out over the cluster, but how do you know where all these processes are? Horde.Registry is the answer. We want Horde.Registry to be above Horde.DynamicSupervisor in the start-up order.
This is what our example above looks like with Horde.Registry added in:
defmodule HelloWorld.Application do
  use Application

  def start(_type, _args) do
    children = [
      {Horde.Registry, [name: HelloWorld.HelloRegistry, keys: :unique]},
      {Horde.DynamicSupervisor, [name: HelloWorld.HelloSupervisor, strategy: :one_for_one]},
    ]

    opts = [strategy: :one_for_one, name: HelloWorld.Supervisor]
    Supervisor.start_link(children, opts)
  end
end
There are also some additional options for Horde.Registry:
	:members, a list of members (if your cluster will have static membership)
	:delta_crdt_options, for tuning the delta CRDT that underpins Horde

See the documentation for Horde.Registry for more information.

  
    
  
  Running a GenServer


Let's create a very simple GenServer that we will use to run in our example application to test things out a little.
defmodule HelloWorld.SayHello do
  use GenServer
  require Logger

  def child_spec(opts) do
    name = Keyword.get(opts, :name, __MODULE__)

    %{
      id: "#{__MODULE__}_#{name}",
      start: {__MODULE__, :start_link, [name]},
      shutdown: 10_000,
      restart: :transient
    }
  end

  def start_link(name) do
    case GenServer.start_link(__MODULE__, [], name: via_tuple(name)) do
      {:ok, pid} ->
        {:ok, pid}

      {:error, {:already_started, pid}} ->
        Logger.info("already started at #{inspect(pid)}, returning :ignore")
        :ignore
    end
  end

  def init(_args) do
    {:ok, nil}
  end

  def via_tuple(name), do: {:via, Horde.Registry, {HelloWorld.HelloRegistry, name}}
end
This GenServer can be started by running the following line of code: Horde.DynamicSupervisor.start_child(HelloWorld.HelloSupervisor, HelloWorld.SayHello).
Once running, you can address this GenServer with the following line of code: GenServer.call(via_tuple(HelloWorld.SayHello), msg).

  
    
  
  Next Steps


Now you should have a working installation of Horde. There is more information available in the other guides, so don't forget to read those too. If you get stuck or have suggestions on how this guide could be improved, please open an issue on Github.



  

    
Eventual Consistency
    

Horde uses a CRDT to sync data between nodes. This means that two nodes in your cluster can have a different view of the data, with differences being merged as the nodes sync with each other. We call this "eventually consistent", and the result is that we have to deal with merge conflicts and race conditions. Horde's CRDT automatically resolves conflicts, but we still have to deal with the after-effects.

  
    
  
  Horde.DynamicSupervisor merge conflict


It is unlikely, but possible, that Horde.DynamicSupervisor will start the same process on two separate nodes.
This can happen:
	if using a custom distribution strategy, or
	when a node dies and not all nodes have the same view of the cluster, or
	if there is a network partition.

Once a network partition has healed, Horde will automatically terminate any duplicate processes.

  
    
  
  Horde.Registry merge conflict


When processes on two different nodes have claimed the same name, this will generate a conflict in Horde.Registry. The CRDT resolves the conflict and Horde.Registry sends an exit signal to the process that lost the conflict. This can be a common occurrence.
Unless this message is handled, it will cause the process to exit. Handling the exit message isn't strictly necessary, because we usually want the process to exit in this case. If for some reason you want to handle the message, simply trap exits in the init/1 callback and handle the message as follows:
def init(arg) do
  Process.flag(:trap_exit, true)
  {:ok, state}
end

def handle_info({:EXIT, _from, {:name_conflict, {key, value}, registry, pid}}, state) do
  # handle the message, add some logging perhaps, and probably stop the GenServer.
  {:stop, :normal, state}
end
Note that, unless your process has restart: :transient in its child spec and you have handled the message to shut down the process cleanly, it will be restarted by its supervisor.
Upon restart, it will try to register itself. This will of course fail. If using a via tuple, the following approach is necessary.
def start_link(arg) do
  case GenServer.start_link(...) do
    {:ok, pid} ->
      {:ok, pid}
    {:error, {:already_started, pid}} ->
      :ignore
  end
end
If you are using Horde.Registry.register/3 in init/1, then you must handle {:error, {:already_registered, pid}}.
def init(arg) do
  case Horde.Registry.register(:my_registry, "key", "value") do
    {:ok, _pid} ->
      {:ok, arg}
    {:error, {:already_registered, _pid}} ->
      :ignore
  end
end



  

    
State Handoff
    

During deployment, when a server is restarted, its processes are restarted on other nodes. It can be useful to hand off state from the process that is shutting down to the process that is starting up.
Horde does not offer a specialized state handoff feature, instead we will be using OTP to accomplish state handoff.

  
    
  
  Trapping exits


When Horde.DynamicSupervisor is shutting down, it will send an exit signal to its child processes. We need to trap this signal to ensure that our terminate/2 callback gets called.
def init(arg) do
  Process.flag(:trap_exit, true)
  {:ok, arg}
end

  
    
  
  Saving state


We can then save the state in the terminate function.
def terminate(reason, state) do
  save_state(state) # save state to Redis, DeltaCRDT, Postgres, Mysql, etc.
end

  
    
  
  Shutdown timeout


The default shutdown timeout is 0, which of course does not leave enough time to save our state. Increase the value of shutdown to make sure that your process has enough time to save its data. Be aware: shutdown must be set at all levels of your application to apply. A more restrictive shutdown timeout at a higher level in the supervision tree will override less restrictive values.
children = [{Horde.DynamicSupervisor, [name: :my_supervisor, shutdown: 1000, strategy: :one_for_one]}]

  
    
  
  Restoring state


Now we have to modify the init/1 callback to restore the state when the process is initializing. We put this in a handle_continue/2 callback to avoid making the supervisor wait unnecessarily for the state to load.
def init(arg) do
  Process.flag(:trap_exit, true)
  {:ok, arg, {:continue, :load_state}}
end

def handle_continue(:load_state, arg) do
  {:noreply, load_state(arg)}
end

  
    
  
  !!WARNING!!


We must be careful not to accidentally load invalid state into our processes. For example, if the state of a process changes between deploys, then you might load invalid state and cause your process to crash. Here are some other things to look out for:
	loading stale data
	loading data fails, causing process to crash
	there is no data to load, causing process to crash

Anytime we are loading state into our processes from an external source, we should be very careful. Erlang's "let it crash" philosophy is predicated on the idea that a process will be in a known good state after a restart, and if we subvert this by loading invalid state, it could have a negative impact on the stability of our system.



  

    
Setting up a Cluster
    

Horde doesn't provide functionality to set up your cluster, we recommend you use libcluster for this purpose.
There are three strategies you can use to integrate libcluster with Horde:

  
    
  
  Automatic Cluster Membership


When starting a Horde.Registry or Horde.DynamicSupervisor, setting the members
option to have a value of :auto will automate membership management. In this
mode, all visible nodes will be initially added to the cluster. In addition,
any new nodes that become visible will be automatically added and any
nodes that shut down will be automatically removed.

  
    
  
  Static Cluster Membership


If you will not be adding or removing members from the cluster dynamically, then you can set up libcluster and tell Horde about the members of your cluster. For example, if you run your cluster on bare metal hardware and have a fixed number of servers.
supervisor_members = [
  {MyHordeSupervisor, :node1},
  {MyHordeSupervisor, :node2},
  {MyHordeSupervisor, :node3},
  {MyHordeSupervisor, :node4}
]

registry_members = [
  {MyHordeRegistry, :node1},
  {MyHordeRegistry, :node2},
  {MyHordeRegistry, :node3},
  {MyHordeRegistry, :node4}
]

children = [
  {Horde.Registry, name: MyHordeRegistry, keys: :unique, members: registry_members},
  {Horde.DynamicSupervisor, name: MyHordeSupervisor, strategy: :one_for_one, members: supervisor_members},
  ...
]
This is the simplest approach. You tell Horde which members are supposed to be in the cluster, and if they are available, Horde will include them in the cluster.

  
    
  
  Dynamic Cluster Membership


If you will be adding and removing nodes from your cluster constantly, and don't want to repackage your application every time you do this, then you will need to perform a couple of extra steps (assuming your needs cannot be met by the :auto setting).
In this scenario, you will need to implement a module-based Supervisor
defmodule MyHordeSupervisor do
  use Horde.DynamicSupervisor

  def start_link(_) do
    Horde.DynamicSupervisor.start_link(
      __MODULE__,
      [strategy: :one_for_one],
      name: __MODULE__
    )
  end

  def init(init_arg) do
    [members: members()]
    |> Keyword.merge(init_arg)
    |> Horde.DynamicSupervisor.init()
  end

  defp members() do
    [Node.self() | Node.list()]
    |> Enum.map(fn node -> {__MODULE__, node} end)
  end
end
Now every time MyHordeSupervisor gets started or restarted, it will compute the members based on the currently connected members.
In this scenario, you may also want to implement a module-based Registry
defmodule MyHordeRegistry do
  use Horde.Registry

  def start_link(_) do
    Horde.Registry.start_link(__MODULE__, [keys: :unique], name: __MODULE__)
  end

  def init(init_arg) do
    [members: members()]
    |> Keyword.merge(init_arg)
    |> Horde.Registry.init()
  end

  defp members() do
    [Node.self() | Node.list()]
    |> Enum.map(fn node -> {__MODULE__, node} end)
  end
end
Now every time MyHordeRegistry gets started or restarted, it will compute the members based on the currently connected members.
We also need a separate process that will listen for {:nodeup, node} and {:nodedown, node} events and adjust the members of the Horde cluster accordingly. Put this in your supervision tree underneath MyHordeSupervisor.
defmodule NodeListener do
  use GenServer

  def start_link(_), do: GenServer.start_link(__MODULE__, [])

  def init(_) do
    :net_kernel.monitor_nodes(true, node_type: :visible)
    {:ok, nil}
  end

  def handle_info({:nodeup, _node, _node_type}, state) do
    set_members(MyHordeRegistry)
    set_members(MyHordeSupervisor)
    {:noreply, state}
  end

  def handle_info({:nodedown, _node, _node_type}, state) do
    set_members(MyHordeRegistry)
    set_members(MyHordeSupervisor)
    {:noreply, state}
  end

  defp set_members(name) do
    members =
      [Node.self() | Node.list()]
      |> Enum.map(fn node -> {name, node} end)

    :ok = Horde.Cluster.set_members(name, members)
  end
end
Note that the functionality provided in this example is essentially the same as
the members: :auto setting, however setting
it up yourself allows greater flexibility to modify it if :auto mode doesn't
meet your requirements.



  

    
Horde.Cluster 
    



      
Public functions to join and leave hordes.
Calling Horde.Cluster.set_members/2 will join the given members in a cluster. Cluster membership is propagated via a CRDT, so setting it once on a single node is sufficient.
{:ok, sup1} = Horde.DynamicSupervisor.start_link([], name: :supervisor_1, strategy: :one_for_one)
{:ok, sup2} = Horde.DynamicSupervisor.start_link([], name: :supervisor_2, strategy: :one_for_one)
{:ok, sup3} = Horde.DynamicSupervisor.start_link([], name: :supervisor_3, strategy: :one_for_one)

:ok = Horde.Cluster.set_members(:supervisor_1, [:supervisor_1, :supervisor_2, :supervisor_3])

Horde.DistributionStrategy behaviour
    



      
Define your own distribution strategy by implementing this behaviour and configuring Horde to use it.
A few distribution strategies are included in Horde, namely:
	Horde.UniformDistribution
	Horde.UniformQuorumDistribution
	Horde.UniformRandomDistribution

Horde.DynamicSupervisor behaviour
    



      
A distributed supervisor.
Horde.DynamicSupervisor implements a distributed DynamicSupervisor backed by a add-wins last-write-wins δ-CRDT (provided by DeltaCrdt.AWLWWMap). This CRDT is used for both tracking membership of the cluster and tracking supervised processes.
Using CRDTs guarantees that the distributed, shared state will eventually converge. It also means that Horde.DynamicSupervisor is eventually-consistent, and is optimized for availability and partition tolerance. This can result in temporary inconsistencies under certain conditions (when cluster membership is changing, for example).
Cluster membership is managed with Horde.Cluster. Joining a cluster can be done with Horde.Cluster.set_members/2. To take a node out of the cluster, call Horde.Cluster.set_members/2 without that node in the list. Alternatively, setting the members startup option to :auto will make Horde auto-manage cluster membership so that all (and only) visible nodes are members of the cluster.
Each Horde.DynamicSupervisor node wraps its own local instance of DynamicSupervisor. Horde.DynamicSupervisor.start_child/2 (for example) delegates to the local instance of DynamicSupervisor to actually start and monitor the child. The child spec is also written into the processes CRDT, along with a reference to the node on which it is running. When there is an update to the processes CRDT, Horde makes a comparison and corrects any inconsistencies (for example, if a conflict has been resolved and there is a process that no longer should be running on its node, it will kill that process and remove it from the local supervisor). So while most functions map 1:1 to the equivalent DynamicSupervisor functions, the eventually consistent nature of Horde requires extra behaviour not present in DynamicSupervisor.

  
    
  
  Divergence from standard DynamicSupervisor behaviour


While Horde wraps DynamicSupervisor, it does keep track of processes by the id in the child specification. This is a divergence from the behaviour of DynamicSupervisor, which ignores ids altogether. Using DynamicSupervisor is useful for its shutdown behaviour (it shuts down all child processes simultaneously, unlike Supervisor).

  
    
  
  Graceful shutdown


When a node is stopped (either manually or by calling :init.stop), Horde restarts the child processes of the stopped node on another node. The state of child processes is not preserved, they are simply restarted.
To implement graceful shutdown of worker processes, a few extra steps are necessary.
	Trap exits. Running Process.flag(:trap_exit) in the init/1 callback of any worker processes will convert exit signals to messages and allow running terminate/2 callbacks. It is also important to include the shutdown option in your child spec (the default is 5000ms).

	Use :init.stop() to shut down your node. How you accomplish this is up to you, but by simply calling :init.stop() somewhere, graceful shutdown will be triggered.



  
    
  
  Module-based Supervisor


Horde supports module-based supervisors to enable dynamic runtime configuration.
defmodule MySupervisor do
  use Horde.DynamicSupervisor

  def start_link(init_arg, options \ []) do
    Horde.DynamicSupervisor.start_link(__MODULE__, init_arg, options)
  end

  def init(init_arg) do
    [strategy: :one_for_one, members: members()]
    |> Keyword.merge(init_arg)
    |> Horde.DynamicSupervisor.init()
  end

  defp members() do
    []
  end
end
Then you can use MySupervisor.child_spec/1 and MySupervisor.start_link/1 in the same way as you'd use Horde.DynamicSupervisor.child_spec/1 and Horde.DynamicSupervisor.start_link/1.

    Functions
  


    
      
        child_spec(init_arg)

      


        See start_link/2 for options.



    


    
      
        count_children(supervisor)

      


        Works like DynamicSupervisor.count_children/1.



    


    
      
        init(options)

      


        Works like DynamicSupervisor.init/1.



    


    
      
        start_child(supervisor, child_spec)

      


        Works like DynamicSupervisor.start_child/2.



    


    
      
        start_link(options)

      


        Works like DynamicSupervisor.start_link/1. Extra options are documented here



    


    
      
        start_link(mod, init_arg, opts \\ [])

      


    


    
      
        stop(supervisor, reason \\ :normal, timeout \\ :infinity)

      


        Works like DynamicSupervisor.stop/3.



    


    
      
        terminate_child(supervisor, child_pid)

      


        Terminate a child process.



    


    
      
        wait_for_quorum(horde, timeout)

      


        Waits for Horde.DynamicSupervisor to have quorum.



    


    
      
        which_children(supervisor)

      


        Works like DynamicSupervisor.which_children/1.

      Link to this type
    
    option()


      
       
       View Source
     


  


  

      

          @type option() ::
  {:name, name :: atom()}
  | {:strategy, Supervisor.strategy()}
  | {:max_restarts, integer()}
  | {:max_seconds, integer()}
  | {:extra_arguments, [term()]}
  | {:distribution_strategy, Horde.DistributionStrategy.t()}
  | {:proxy_message_ttl, integer() | :infinity}
  | {:shutdown, integer()}
  | {:members, [Horde.Cluster.member()] | :auto}
  | {:delta_crdt_options, [DeltaCrdt.crdt_option()]}
  | {:process_redistribution, :active | :passive}


      



  



  
    
      
      Link to this type
    
    options()

          @type options() :: [option()]

      Link to this callback
    
    child_spec(options)

          @callback child_spec(options :: options()) :: Supervisor.child_spec()


      



  



  
    
      
      Link to this callback
    
    init(options)

          @callback init(options()) :: {:ok, options()} | :ignore

      Link to this function
    
    child_spec(init_arg)


      
       
       View Source
     


  


  

See start_link/2 for options.

  



  
    
      
      Link to this function
    
    count_children(supervisor)

Works like DynamicSupervisor.count_children/1.
This function delegates to all supervisors in the cluster and returns the aggregated output.

  



  
    
      
      Link to this function
    
    init(options)

Works like DynamicSupervisor.init/1.

  



  
    
      
      Link to this function
    
    start_child(supervisor, child_spec)

Works like DynamicSupervisor.start_child/2.

  



  
    
      
      Link to this function
    
    start_link(options)


      
       
       View Source
     


  


  

Works like DynamicSupervisor.start_link/1. Extra options are documented here:
	:distribution_strategy, defaults to Horde.UniformDistribution, but more are available - see Horde.DistributionStrategy
	:proxy_message_ttl, defaults to :infinity. Can be set to an integer indicating the maximum number of times a message may be forwarded in a Horde.DynamicSupervisor cluster. Leaving it at infinity is genrally fine when using a stable distribution strategy such as Horde.UniformDistribution. Setting a TTL is helpful when migrating to a different distribution_strategy, or when using an algorithm with random distribution such as Horde.UniformRandomDistribution, as it will prevent messages from looping (near) infinitely.

    stop(supervisor, reason \\ :normal, timeout \\ :infinity)

Works like DynamicSupervisor.stop/3.

  



  
    
      
      Link to this function
    
    terminate_child(supervisor, child_pid)

          @spec terminate_child(Supervisor.supervisor(), child_pid :: pid()) ::
  :ok
  | {:error, :not_found}
  | {:error, {:node_dead_or_shutting_down, String.t()}}


      


Terminate a child process.
Works like DynamicSupervisor.terminate_child/2.

  



  
    
      
      Link to this function
    
    wait_for_quorum(horde, timeout)

          @spec wait_for_quorum(horde :: GenServer.server(), timeout :: timeout()) :: :ok


      


Waits for Horde.DynamicSupervisor to have quorum.

  



  
    
      
      Link to this function
    
    which_children(supervisor)

Works like DynamicSupervisor.which_children/1.
This function delegates to all supervisors in the cluster and returns the aggregated output. Where memory warnings apply to DynamicSupervisor.which_children, these count double for Horde.DynamicSupervisor.which_children.

          @type status() :: :uninitialized | :alive | :shutting_down | :dead

          @type t() :: %Horde.DynamicSupervisor.Member{name: term(), status: term()}


      



  


        

      



  

    
Horde.NodeListener 
    



      
A cluster membership manager.
Horde.NodeListener monitors nodes in BEAM's distribution system and
automatically adds and removes those marked as visible from the cluster it's
managing

        Returns a specification to start this module under a supervisor.

Returns a specification to start this module under a supervisor.
See Supervisor.

Callback implementation for GenServer.init/1.

  



  
    
      
      Link to this function
    
    make_members(cluster)


      
       
       View Source
     


  


  

      

          @spec make_members(atom()) :: [{atom(), node()}]

          @spec start_link(atom()) :: GenServer.on_start()


      



  


        

      



  

    
Horde.Registry behaviour
    



      
A distributed process registry.
Horde.Registry implements a distributed Registry backed by a δ-CRDT (provided by DeltaCrdt). This CRDT is used for both tracking membership of the cluster and implementing the registry functionality itself. Local changes to the registry will automatically be synced to other nodes in the cluster.
Cluster membership is managed with Horde.Cluster. Joining a cluster can be done with Horde.Cluster.set_members/2. To take a node out of the cluster, call Horde.Cluster.set_members/2 without that node in the list. Alternatively, setting the members startup option to :auto will make Horde auto-manage cluster membership so that all (and only) visible nodes are members of the cluster.
Horde.Registry supports the common "via tuple", described in the documentation for GenServer.
Horde.Registry is API-compatible with Registry, with the following exceptions:
	Horde.Registry does not support keys: :duplicate.
	Horde.Registry does not support partitions.
	Horde.Registry sends an exit signal to a process when it has lost a naming conflict. See Horde.Registry.register/3 for details.


  
    
  
  Module-based Registry


Horde supports module-based registries to enable dynamic runtime configuration.
defmodule MyRegistry do
  use Horde.Registry

  def start_link(_) do
    Horde.Registry.start_link(__MODULE__, [keys: :unique], name: __MODULE__)
  end

  def init(init_arg) do
    [members: members()]
    |> Keyword.merge(init_arg)
    |> Horde.Registry.init()
  end

  defp members() do
    [Node.self() | Node.list()]
    |> Enum.map(fn node -> {__MODULE__, node} end)
  end
end
Then you can use MyRegistry.child_spec/1 and MyRegistry.start_link/1 in the same way as you'd use Horde.Registry.child_spec/1 and Horde.Registry.start_link/1.

        See Registry.count/1.

        See Registry.count_match/4.

        See Registry.delete_meta/2.



    


    
      
        dispatch(registry, key, mfa_or_fun, opts \\ [])

      


        See Registry.dispatch/4.

        Works like Registry.init/1.

        See Registry.keys/2.

        See Registry.put_meta/3.

        Register a process under the given name. See Registry.register/3.

        See Registry.select/2.

        See Registry.unregister/2.

        See Registry.unregister_match/4.



    


    
      
        update_value(registry, key, callback)

      


        See Registry.update_value/3.

See start_link/2 for options.

          @spec count(registry :: Registry.registry()) :: non_neg_integer() | :undefined


      


See Registry.count/1.

  



    

  
    
      
      Link to this function
    
    count_match(registry, key, pattern, guards \\ [])


      
       
       View Source
     


  


  

See Registry.count_match/4.

          @spec delete_meta(registry :: Registry.registry(), name :: Registry.key()) :: :ok


      


See Registry.delete_meta/2.

See Registry.dispatch/4.

    init(options)


      
       
       View Source
     


  


  

Works like Registry.init/1.

          @spec keys(registry :: Registry.registry(), pid()) :: [Registry.key()]


      


See Registry.keys/2.

See Registry.lookup/2.

See Registry.match/4.

          @spec meta(registry :: Registry.registry(), key :: Registry.meta_key()) ::
  {:ok, Registry.meta_value()} | :error


      


See Registry.meta/2.

  registry :: Registry.registry(),
  key :: Registry.meta_key(),
  value :: Registry.meta_value()
) :: :ok


      


See Registry.put_meta/3.

          @spec register(
  registry :: Registry.registry(),
  name :: Registry.key(),
  value :: Registry.value()
) :: {:ok, pid()} | {:error, {:already_registered, pid()}}


      


Register a process under the given name. See Registry.register/3.
When 2 clustered registries register the same name at exactly the
same time, it will seem like name registration succeeds for both
registries. The function returns {:ok, pid} for both of these
calls.
However, due to the eventually consistent nature of the CRDT,
conflict resolution will take place, and the CRDT will pick one of
the two processes as the "winner" of the name. The losing process
will be sent an exit signal (using Process.exit/2) with the
following reason:
{:name_conflict, {name, value}, registry_name, winning_pid}
When two registries are joined using Horde.Cluster.set_members/2,
this name conflict message can also occur.
When a cluster is recovering from a netsplit, this name conflict
message can also occur.

See Registry.start_link/1.
Does not accept [partitions: x], nor [keys: :duplicate] as options.

See Registry.unregister/2.

See Registry.unregister_match/4.

See Registry.update_value/3.

Distributes processes to nodes uniformly using a hash ring.
Given the same set of members, it will always start
the same process on the same node.

Distributes processes to nodes uniformly using a hash ring. Contains a quorum mechanism to handle netsplits.
It enforces a quorum and will shut down all processes on a node if it is split from the rest of the cluster.

Distributes processes with an uniform probability to
any node that is alive.

